



vertical well

$$\pi r_e^2 = A = 500 \cdot 500$$

$$r_e = \sqrt{\frac{500 \cdot 500}{\pi}}$$

Reservoir top area	[m <sup>2</sup> ]	2.50E+05
Reservoir pressure, p <sub>R</sub>	[bara]	300
Flowing bottom-hole pressure, p <sub>wf</sub>	[bara]	200
p <sub>av</sub>	[bara]	250
Oil viscosity, μ <sub>o</sub> at average pressure	[cp]	1.877
Oil volume factor, B <sub>o</sub> , at average pressure	[m <sup>3</sup> /Sm <sup>3</sup> ]	1.144
Wellbore radius, r <sub>w</sub>	[m]	0.15
<b>Vertical well located in the center and perforated throughout</b>		
External radius, r <sub>e</sub>	[m]	282.1
Skin, s	[ <sup>-</sup> ]	0
Shape factor, s <sub>A</sub>	[ <sup>-</sup> ]	0.012
Productivity Index, J	[Sm <sup>3</sup> /d/bar]	14.7
<b>Horizontal well</b>		
Wellbore length	[m]	500
Elevation difference between toe and heel, b (sign doesn't matter)	[m]	0
Productivity Index, J	[Sm <sup>3</sup> /d/bar]	63.2

```

Function J_vertical(k, h, Uo, Bo, re, rw, s, sa)
    'Productivity index for vertical well, undersaturated oil, pss, in Sm3/d/bar
    'k in md
    'h in m
    'Uo in cp
    'Bo in m^3/Sm^3
    'fa is shape factor
    J_vertical = (k * h) / (18.68 * Uo * Bo * (Log(re / rw) - 0.75 + s + sa))
    'Natural Log in Visual Basic is Log, not LN'
End Function

Function J_horizontal(L, D, h, b, Lw, kh, kv, Bo, viso, rw)
    'Productivity index for horizontal well, undersaturated oil, pss, in Sm3/d/bar
    'L Reservoir length along well direction [m]
    'D reservoir width [m]
    'h reservoir thickness [m]
    'Lw well length [m]
    'kh horizontal permeability [md]
    'kv vertical permeability [md]
    'Bo oil formation volume factor [m^3/Sm^3]
    'viso oil viscosity [cp]
    'rw well radius [m]
    'b, height difference between heel and toe [m]
    Pi = Atan(1) * 4
    b = Abs(b)
    If b / h > 0.1 Then
        s_b = 0.69
    Else
        s_b = 0
    End If
    beta = (kh / kv) ^ 0.5
    Lw_hat = Lw * (1 + ((b / Lw) ^ 2) * (beta ^ 2 - 1)) ^ 0.5
    Lw_bar = Lw * (1 - (b / Lw) ^ 2) ^ 0.5
    rw_hat = 0.5 * rw * (1 + (1 + (beta ^ 2 - 1) * ((Lw_bar / Lw) ^ 2)) ^ 0.5)
    A1 = 0.53 * ((L / D) ^ 2) + 1.15 * (L / D) + 0.164
    A2 = (1 - (Lw_bar / L)) / (0.45 + (Lw_bar / L))
    fa = (Lw_bar / L) * (1 + A1 * A2)
    C1 = 3 * h * beta * (Log(beta * h / (2 * Pi * rw_hat)) + s_b) / Lw_hat
    C2 = (Pi * D * fa / (2 * Lw_bar))
    unit_conversion_constant = (9.869E-13 * 0.001) * 24 * 3600 * 100000 * 6 * Pi / (0.001)
    J_horizontal = unit_conversion_constant * kh * h / (viso * Bo * (C1 + C2))
End Function

```